



# Beetle-Bot

---

C. H. Ting

December 15, 2018

SVFIG




# Summary

---

- **Beetle-Bot**
- **PWM Motor drivers**
- **ADC Sensors**
- **IR Remote Sensor**
- **Speakers**
- **Demo**

# Beetle-Bot





# PWM Motor Driver

---

```
: init-ports
    6 24 c! 60 2a c! \ output ports
    a3 44 c! 5 45 c! \ TCCR0A, TCCR0B
    a1 80 c! d 81 c! \ TCCR1A, TCCR1B ;
: s1 ( n -- ) 88 c! ; : s2 ( n -- ) 8a c! ;
: s5 ( n -- ) 47 c! ; : s6 ( n -- ) 48 c! ;
: stop 0 s1 0 s2 0 s5 0 s6 ;
: go FF s1 0 s2 FF s5 0 s6 ;
: back 0 s1 FF s2 0 s5 FF s6 ;
: left C0 s1 0 s2 0 s5 0 s6 ;
: right 0 s1 0 s2 C0 s5 0 s6 ;
```



# Direct Motor Drivers

---

```
: init-ports 26 24 c! 60 2a c! ;  
: stop 0 25 c! 0 2b c! ;  
: back 4 25 c! 20 2b c! ;  
: right 0 25 c! 40 2b c! ;  
: left 2 25 c! 0 2b c! ;  
: go 2 25 c! 40 2b c! ;
```



# ADC Sensors

---

```
: adc ( port -- n )
  40 or 7c c! c3 7a c! 1 ms 78 @ ;
: test adc . ;
: tests 0 test 1 test 2 test 3 test 4 test ;
: demo2
begin
  3 adc 100 < 4 adc 100 <
  if if back else right then
  else if left else go then
  then 10 ms
?key until drop stop ;
```




# Digital Sensors

---

```
: sensors ( -- n ) 26 c@ ;
: demo4
  begin cr sensors . 100 ms ?key until drop ;
: drive sensors 18 and
  dup 10 = if right drop exit then
  dup 8 = if left drop exit then
    18 = if go else back then ;
: demo5
  begin drive 10 ms ?key until
  drop stop ;
```

# IR Remote Signals







# IR Remote Signal Timing

---

PWM ON	OFF
2.4 ms	0.6 ms
1.2 ms	0.6 ms
0.6 ms	0.6 ms
1.2 ms	0.6 ms
0.6 ms	0.6 ms
1.2 ms	0.6 ms
0.6 ms	0.6 ms
0.6 ms	0.6 ms
1.2 ms	0.6 ms
0.6 ms	0.6 ms
0.6 ms	0.6 ms
0.6 ms	0.6 ms
0.6 ms	270 ms



# IR Remote Decoder

---

```
: low ( -- ) begin 23 c@ 10 and while  
repeat ;  
: high ( -- ) begin 23 c@ 10 and until ;  
: bit-width ( -- n ) high 0 ( n )  
begin 23 c@ 10 and while 1+ repeat ;  
: start-bit low  
begin bit-width 180 280 within until ;  
: get-word ( -- n )  
0 F for 2* bit-width 80 > if 1 or then  
next ;  
: get-code ( -- d )  
start-bit get-word get-word 100 ms ;
```



# IR Remote Controller

---

```
5AA5 constant '>'    4AB5 constant 'v'  
10EF constant '<'    18E7 constant '^'  
38C7 constant '0'
```

```
: beetle init-ports  
begin get-code cr dup .  
  dup '>' = if right then  
  dup '<' = if left then  
  dup '^' = if go then  
  dup 'v' = if back then  
  '0' = if stop then drop  
?key until drop stop ;
```



# Speakers

---

: `init-ports`

```
0a 24 c! 40 2a c!      \ output ports
42 44 c!  0 45 c!      \ TCCR0A, TCCR0B
40 80 c!  0 81 c!      \ TCCR1A, TCCR1B
42 0b0 c!  0 0b1 c! ; \ TCCR2A, TCCR2B
```

: `ch0 (note scaler -- )`

```
0 45 c! 0 46 c!  swap 47 c! 45 c! ;
```

: `ch1 (note scaler -- )`

```
0 81 c! 0 84 c!  swap 88 c! 81 c! ;
```

: `ch2 (note scaler -- )`

```
0 b1 c! 0 b2 c!  swap 0b3 c! 0b1 c! ;
```



# Interrupt Vectors

---

```
: INIT-VECTORS
  4 18 FOR
    940C ( jmp ) OVER I! 2+
    3F ( i-return ) OVER I! 2+
  NEXT DROP
  9518 7E ( i-return ) I! ;
: INTERRUPT ( addr vector -- )
  SWAP 2/ SWAP
  1- 2* 1+ 2* I! ;
```



# Assembler

---

```
: RETI, 9518 ( RETI ) ,  
  [COMPILE] [ OVERT ; IMMEDIATE  
: SEI, 9478 ( SEI ) , ; IMMEDIATE  
: CLI, 94F8 ( CLI ) , ; IMMEDIATE  
: NOP, 0 ( NOP ) , ; IMMEDIATE  
: SBI, 9A1D ( SBI PINB,5 ) , ; IMMEDIATE  
: PUSH-SREG,  
  B68F ( IN r8,sreg ) ,  
  928F ( PUSH r8 ) , ; IMMEDIATE  
: POP-SREG,  
  908F ( POP r8 ) ,  
  BE8F ( OUT sreg,r8 ) , ; IMMEDIATE
```



# Watch Dog Timer

---

```
: ENABLE-WDT ( 184X -- )
  [ 94F8 , ( CLI )
    95A8 , ( WDR )
    9390 , 60 , ( STS CSR,tosh )
    9380 , 60 , ( STS CSR,tosl )
    9478 , ( SEI )
  ] ; ( RET )
: WDT
  PUSH-SREG,
  SBI, ( SBI PINB,5 flip D13 output )
  POP-SREG,
  RETI,
```



# Beetle-Bot Demo

---


```
: beetle init-ports
begin get-code cr dup .
  dup '>' = if right then
  dup '<' = if left then
  dup '^' = if go then
  dup 'v' = if back then
  '0' = if stop then drop
?key until drop stop ;
```





---

Questions?



---

Thank you.